時間発展する問題を解く

桂田 祐史
katurada AT meiji.ac.jp

2015 年 12 月 8 日

1　前回の講義のメモ: 熱方程式に対する有限要素法

1.1　準備 — Poisson 方程式を思い出す

まず Poisson 方程式の弱形式とプログラム例を示す。

$$\Omega = (0,1)\times(0,1), \quad \Gamma := \partial\Omega,$$
$$\Gamma_1 = \{(0,y) \mid 0 \le y \le 1\} \cup \{(x,0) \mid 0 \le x \le 1\}, \quad \Gamma_2 := \Gamma \setminus \Gamma_1.$$

$f\colon \Omega \to \mathbb{R}$, $g_1\colon \Gamma_1 \to \mathbb{R}$, $g_2\colon \Gamma_2 \to \mathbb{R}$ が与えられたとして、次の境界値問題を考える。

$$\begin{aligned} &-\triangle u(x,y) = f(x,y) \quad ((x,y)\in\Omega),\\ &u(x,y) = g_1(x,y) \quad ((x,y)\in\Gamma_1),\\ &\frac{\partial u}{\partial n}(x,y) = g_2(x,y) \quad ((x,y)\in\Gamma_2). \end{aligned}$$

Find $u \in X_{g_1}$ s.t.

$$\langle u, v\rangle - (f,v) - [g_2, v] = 0 \quad (v \in X).$$

ここで

$$X_{g_1} := \left\{ w \in H^1(\Omega) \;\middle|\; w = g_1 \quad \text{on } \Gamma_1 \right\}, \quad X := \left\{ v \in H^1(\Omega) \;\middle|\; v = 0 \quad \text{on } \Gamma_1 \right\}.$$

poisson.edp

```
// 菊地文雄，有限要素法概説，サイエンス社
int m=10;
mesh Th=square(m,m);
plot(Th,wait=true);
fespace Vh(Th,P1);
Vh u,v;
func f=1;
func g1=0;
func g2=0;
solve Poisson(u,v)=
  int2d(Th)(dx(u)*dx(v)+dy(u)*dy(v))
 -int2d(Th)(f*v)
 -int1d(Th,2,3)(g2*v)
 +on(1,4,u=g1);
plot(u,wait=1,ps="kikuchi.ps");
```

square() は辺が座標軸に平行な長方形領域を三角形分割する関数である。下の辺、右の辺、上の辺、左の辺の順に 1, 2, 3, 4 というラベルをつける。

int1d(Th,2,3) はラベル 2, 3 の上の線積分である。

自分で境界を定義してから、囲まれる領域を分割するプログラムは、「FreeFem++ の紹介」[1] を見よ。

1.2 熱方程式に対する 後退 Euler 法

$u^n = u(\cdot, t_n)$ とするとき、u_t を後退差分近似して

$$\frac{1}{\tau}\left(u^n - u^{n-1}, v\right) + \langle u^n, v\rangle - (f, v) - [g_2, v] = 0.$$

あるいは

$$\left(u^{n+1} - u^n, v\right) + \tau\langle u^{n+1}, v\rangle - \tau(f, v) - \tau[g_2, v] = 0.$$

(f, g_2 が時間依存している場合は、f^{n+1}, g_2^{n+1} のようにすべきである。)

[1] http://nalab.mind.meiji.ac.jp/~mk/labo/text/welcome-to-freefem-2012/node8.html

heatB.edp

```
// 菊地文雄，有限要素法概説，サイエンス社の Poisson 方程式の問題の非定常版
int m=10;
real Tmax=10, tau=0.01, t;
func f=1;
func g1=0;
func g2=0;
func u0=sin(pi*x)*sin(pi*y);
mesh Th=square(m,m);
plot(Th,wait=true);
fespace Vh(Th,P1);
Vh u=u0,uold,v;
problem heat(u,v)=
  int2d(Th)(u*v+tau*(dx(u)*dx(v)+dy(u)*dy(v)))
 -int2d(Th)(tau*f*v)
 -int1d(Th,2,3)(tau*g2*v)
 -int2d(Th)(uold*v)
 +on(1,4,u=g1);
for (real t=0;t<Tmax;t+=tau) {
  uold=u;
  heat;
  plot(u,wait=0);
}
plot(u,wait=1);
```

1.3 熱方程式に対する θ 法

FreeFem++ のマニュアルに §9.5.1 Mathematica Theory on Time Difference Approximations という項があり、θ 法のサンプル・プログラムが載っている。(相変わらず凝った例である。シンプルなのを載せれば良いのに…)

前項と同じ熱方程式の場合に説明する。

$$u^{n+\theta} := \theta u^{n+1} + (1-\theta)u^n$$

とおいて (もし f や g_2 が時間依存する場合は f や、g_2 も同様に処理すべき)、

$$\frac{1}{\tau}\left(u^{n+1}-u^n, v\right) + \langle u^{n+\theta}, v\rangle - (f,v) - [g_2, v] = 0.$$

2 本日の実習

1. 前節の 2 つのプログラムを入力＆実行し、熱方程式版の最終結果 ($t = \mathrm{T}_{\max}$) が、Poisson 方程式の結果とほぼ同じであることを確認せよ。

2. θ 法のプログラム heatT.edp を作成せよ。

3. 安定性について実験的に調べよ。(差分法では、$1/2 \le \theta \le 1$ の場合は無条件安定、$0 \le \theta < 1/2$ の場合は $0 < \lambda \le \dfrac{1}{2(1-2\theta)}$ が安定のための必要十分条件であった。ただし $\lambda = \dfrac{\tau}{h_x^2} + \dfrac{\tau}{h_y^2}$.)

4. (もし出来れば) 厳密解が分かる問題を選び、誤差を調べよ。

FreeFem++ には、マニュアルもあるが、簡単な使い方は、桂田 [?], [?] で分かると思う。
理論の理解には、齊藤 [?] がお勧め。

3 おまけ

波動方程式の初期値境界値問題 (とりあえず太鼓) を解くプログラムを作れ。

(1) $$\frac{1}{c^2}u_{tt} = \triangle u \quad ((x,y,t) \in \Omega \times (0,\infty)),$$

(2) $$u(x,y,t) = 0 \quad ((x,y,t) \in \partial\Omega \times (0,\infty)),$$

(3) $$u(x,y,0) = \phi(x,y), \quad u_t(x,y,0) = \psi(x,y) \quad ((x,y) \in \overline{\Omega}).$$

$$\left(\frac{u^{n+1} - 2u^n + u^{n-1}}{\tau^2}, v\right) = c^2 \langle u^n, v\rangle$$

u^1 については、次を参考にすること。

(1) Taylor 展開で 1 次近似

$$u(x,y,\tau) \fallingdotseq u(x,y,0) + u_t(x,y,0)\tau = \phi(x,y) + \tau\psi(x,y).$$

(2) Taylor 展開で 2 次近似

$$\begin{aligned} u(x,y,\tau) &\fallingdotseq u(x,y,0) + u_t(x,y,0)\tau + \frac{u_{tt}(x,y,0)}{2}\tau^2 \\ &= u(x,y,0) + u_t(x,y,0)\tau + \frac{\tau^2}{2}\cdot c^2 \triangle u(x,y,0) \\ &= \phi(x,y) + \psi(x,y)\tau + \frac{(c\tau)^2}{2}\triangle\phi(x,y). \end{aligned}$$

4 おまけの追加 (授業中にとっさに紹介したプログラム)

差分法の安定性の話をする際に、従来サンプル・プログラムは C+GLSC で記述したものを紹介していたが、そういう環境を持っていない学生がいたので、以前冗談半分に書いた FreeFem++ 言語による、1 次元熱方程式の初期値境界値問題のプログラムを紹介した。

heat1d-e-freefem.edp

```
// heat1d-e-freefem.edp
// 実行: FreeFem++ heat1d-e-freefem.edp
// N, x が大域的な識別子を隠すと警告が出る（つまり名前が衝突する)。

int i, N=50, n, nMax;
real h = 1.0 / N, lambda = 0.5, Tmax = 1.0;
real tau = lambda * h^2;
real[int] x(N+1);
real[int] u(N+1);
real[int] newu(N+1);

cout << "h= " << h << endl;
cout << "lambda= " << lambda << endl;
cout << "tau= " << tau << endl;

func real f(real x) {
  if (x < 0.5)
    return x;
  else
    return 1 - x;
}

for (i = 0; i <= N; i++) {
  x[i] = i * h;
  u[i] = f(x[i]);
}
plot([x,u], bb=[[-0.1,-0.1],[1.1,1.1]],aspectratio=true, wait=true);

nMax = rint(Tmax / tau);
cout << "nMax =" << nMax << endl;
for (n = 1; n <= nMax; n++) {
  for (i = 1; i < N; i++)
    newu[i] = (1.0 - 2.0 * lambda) * u[i] + lambda * (u[i+1] + u[i-1]);
  // cout << newu << endl;
  newu[0] = newu[N] = 0;
  u = newu;
  plot([x, u], bb=[[-0.1,-0.1],[1.1,1.1]],aspectratio=true);
}
```

http://nalab.mind.meiji.ac.jp/~mk/ouyousuchi/heat1d-e-freefem.edp から入手可能である。

誰か、θ 法のプログラム書いてくれないかな。

以下、解説をちびちび書き足していく (これは 2017 年 8 月 28 日に組版したもの)。

5 解説

5.1 θ 法による熱方程式

u_t を後退差分近似して

$$\frac{1}{\tau}\left(u^n - u^{n-1}, v\right) + \langle u^n, v\rangle - (f, v) - [g_2, v] = 0.$$

番号をずらして

$$\frac{1}{\tau}\left(u^{n+1} - u^n, v\right) + \langle u^{n+1}, v\rangle - (f, v) - [g_2, v] = 0. \tag{4}$$

一方、前進差分近似して、

$$\frac{1}{\tau}\left(u^{n+1} - u^n, v\right) + \langle u^n, v\rangle - (f, v) - [g_2, v] = 0. \tag{5}$$

(4) と (5) を $\theta : (1-\theta)$ の比で加重平均を取ると、

$$\frac{1}{\tau}\left(u^n - u^{n-1}, v\right) + \theta\langle u^{n+1}, v\rangle + (1-\theta)\langle u^n, v\rangle - (f, v) - [g_2, v] = 0. \tag{6}$$

$u^{n+\theta} := \theta u^{n+1} + (1-\theta)u^n$ とおくと、確かに

$$\frac{1}{\tau}\left(u^n - u^{n-1}, v\right) + \langle u^{n+\theta}, v\rangle + (1-\theta)\langle u^n, v\rangle - (f, v) - [g_2, v] = 0$$

が得られる。これは記憶用には短くて便利かもしれないが、プログラムを書くには、むしろ (6) の方が便利であろう。

```
// 菊地文雄，有限要素法概説，サイエンス社の Poisson 方程式の非定常版
int m=10;
real Tmax=10, tau=0.01, t, theta=0.5;
func f=1;
func g1=0;
func g2=0;
func u0=sin(pi*x)*sin(pi*y);
mesh Th=square(m,m);
plot(Th,wait=true);
fespace Vh(Th,P1);
Vh u=u0,uold,v;
problem heat(u,v)=
  int2d(Th)(u*v+theta*tau*(dx(u)*dx(v)+dy(u)*dy(v)))
 +int2d(Th)((1-theta)*tau*(dx(uold)*dx(v)+dy(uold)*dy(v)))
 -int2d(Th)(tau*f*v)
 -int1d(Th,2,3)(tau*g2*v)
 -int2d(Th)(uold*v)
 +on(1,4,u=g1);
for (real t=0;t<Tmax;t+=tau) {
  uold=u;
  heat;
  plot(u,wait=0);
}
plot(u,wait=1);
```

プログラム中の m を大きくすると、メッシュ分割が細かくなる。差分法では空間の刻み幅 h_x, h_y を小さくすることを意味する。

m, τ, θ を色々変えて計算を実行する。実行時に変えられると便利かもしれない。

```
  cout << "m τ  θ ";
  cin >> m >> tau >> theta;
```

鈴木厚先生のチュートリアルを聴いて、LU 分解は最初だけやるべきだ！ということに気がついて(恥ずかしや)、以下のプログラムを作った。

```
// heat.edp --- 正方形領域で熱方程式 u_t=u_{xx}+u_{yy}+f を解く
int i,m=100;
real Tmax=1, tau=0.01, t;
func f=1;
func g1=0;
func g2=0;
func u0=sin(pi*x)*sin(pi*y);
mesh Th=square(m,m);
plot(Th,wait=true);
fespace Vh(Th,P1);
//
Vh u=u0,up,v;
problem heat(u,v,solver=UMFPACK,init=i)=
  int2d(Th)(u*v/tau+dx(u)*dx(v)+dy(u)*dy(v))
 -int2d(Th)(f*v)
 -int1d(Th,2,3)(g2*v)
 -int2d(Th)(up*v/tau)
 +on(1,4,u=g1);
for (i=0; i < Tmax/tau; i++) {
  up=u;
  t=(i+1)*tau;
  heat;
  // plot(u,cmm="t="+t,wait=0,ps="heat"+i+".ps");
  plot(u,cmm="t="+t,wait=0);
}
plot(u,cmm="t="+t,wait=1,ps="heat.ps");
```

ループの制御変数を i という変数にして、problem に ,init=i と書き足すのがミソ。最初は i が 0 であるので、init が False であるが、それ以降は $i \neq 0$ であるので、init が True である、つまり LU 分解済みだ、と指示するわけである。

5.2 太鼓の振動

taiko.edp

```
// taiko.edp

border Gamma(t=0,2*pi) { x=cos(t); y=sin(t); }
mesh Th=buildmesh(Gamma(40));
plot(Th,wait=true);

fespace Vh(Th,P1);
Vh newu,u,oldu,v;

func phi=1-x^2-y^2;
func psi=0;

real tau=0.01,Tmax=100;

u=phi;
newu=u+tau*psi;

problem wave(newu,v)=
  int2d(Th)(newu*v)
 +int2d(Th)(-2*u*v+oldu*v+tau^2*(dx(u)*dx(v)+dy(u)*dy(v)))
 +on(Gamma,newu=0);

for (real t=0; t<Tmax; t+=tau) {
    oldu = u;
    u = newu;
    wave;
    plot(u,dim=3,wait=0);
}
plot(u,wait=1);
```

plot で ,dim=3 とした場合、描画範囲どうやって指定するのだろう。それを何とかしたいのだけど、誰か解決策を見つけた人、教えて下さい。

それから、これも ,init=i の技が使えるのかな？これもやってみた人、教えて下さい。